CSN 450 ‑ Computer Graphics ‑ POV image map 
Due, Monday, April 22
You may find it helpful to read the information on using image maps in "Help on the POV Ray Scene Language".   It can be located under the heading "Textures" in the table of contents or by entering “image_maps” in the index of the help file.  
1. Create a pov file named TextureMap.pov which defines two objects (with different underlying shapes) and refines their appearances using image maps.  Use at least one image_map and one bump_map.
2. To begin, choose two small, ordinary, manmade objects (can of soup, dart board, billiard ball, cereal box, etc.) that you can physically bring to class.  

3. Model your objects to create rendered images of these ordinary items.  You may use images that you locate on the Internet or that you create in a package such as Photoshop or Gimp.
Part of the grade will reflect your effort and success in creating realistic looking objects.  Upload the pov file and the required images to Moodle.  If you are using another package to create your scene, upload (or email) the necessary files along with instructions on how to create your rendered image (particularly where image maps must be located).
Some examples follow with excerpts from POV-Ray Help.  For the examples, camera and lighting were set as:

camera {    

    location <0,6,-6>    // Camera location 

    look_at <0, 0, 0>    // pointed at origin
}
light_source { <2, 8, -5> color White }
Using Bump Maps in POV-Ray:

A bumpy surface on an object can be simulated by altering the way light reflects off of the surface. Reflection calculations depend on a vector called a surface normal vector. This is a vector which points away from the surface and is perpendicular to it. By artificially modifying (or perturbing) this normal vector you can simulate bumps. In POV-Ray this is done by adding an optional normal statement. 

Note: that attaching a normal pattern does not really modify the surface. It only affects the way light reflects or refracts at the surface so that it looks bumpy. 

The syntax is: 

NORMAL:

  normal { [NORMAL_IDENTIFIER] [NORMAL_TYPE] [NORMAL_MODIFIER...] }

NORMAL_TYPE:

  PATTERN_TYPE Amount |

  bump_map { BITMAP_TYPE "bitmap.ext" [BUMP_MAP_MODS...]}

NORMAL_MODIFIER:

  PATTERN_MODIFIER | NORMAL_LIST | normal_map { NORMAL_MAP_BODY } |

  slope_map{ SLOPE_MAP_BODY } | bump_size Amount |

  no_bump_scale Bool | accuracy Float

When all else fails and none of the above normal pattern types meets your needs you can use a bump_map to wrap a 2-D bit-mapped bump pattern around your 3-D objects. 

Instead of placing the color of the image on the shape like an image_map a bump_map perturbs the surface normal based on the color of the image at that point. The result looks like the image has been embossed into the surface. By default, a bump map uses the brightness of the actual color of the pixel. Colors are converted to gray scale internally before calculating height. Black is a low spot, white is a high spot. 
The syntax for a bump_map is: 

BUMP_MAP:

    normal

    {

        bump_map

        {

            BITMAP_TYPE "bitmap.ext"

            [BUMP_MAP_MODS...]

        }

        [NORMAL_MODFIERS...]

    }

BITMAP_TYPE:

    gif | tga | iff | ppm | pgm | png | jpeg | tiff | sys

BUMP_MAP_MOD:

    map_type Type | once | interpolate Type | use_color | 

    use_colour | bump_size Value

// bump mapping example
sphere {                                   

    <0, 1, 0>, 2

    texture {

      pigment { color Yellow }

      normal { bumps 0.4 scale 0.1 }   // bumps sets depth, scale sets width
      finish{ ambient 0.1 }

    }

 }
// bump mapping example with bump_map image

sphere{ <0,0,0>, 0.5

        texture{ pigment{ color rgb<1, 1,1>}

                 normal{ 

                   bump_map{

                      jpeg "bitmaps/triangles.jpg" 

                      interpolate 2

                   }  

                   scale 0.25  

                 }

                 finish { diffuse 0.9 phong 0.1}

               } // end of texture 1 

        rotate <45, 0, 0>

        scale 7    

}
Using Image Maps in POV-Ray:

The syntax for an image_map is: 

IMAGE_MAP:

    pigment

    {

        image_map

        {

            [BITMAP_TYPE] "bitmap[.ext]"

            [IMAGE_MAP_MODS...]

        }

        [PIGMENT_MODFIERS...]

    }

BITMAP_TYPE:

    gif | tga | iff | ppm | pgm | png | jpeg | tiff | sys

IMAGE_MAP_MOD:

    map_type Type | once | interpolate Type | 

    filter Palette, Amount | filter all Amount |

    transmit Palette, Amount | transmit all Amount

Filenames specified in the image_map statements will be searched for in the home (current) directory first and, if not found, will then be searched for in directories specified by any +L or Library_Path options active. This would facilitate keeping all your image maps files in a separate subdirectory and giving a Library_Path option to specify where your library of image maps are. 

once
Normally there are an infinite number of repeating image maps, bump maps or material maps created over every unit square of the x-y-plane like tiles. By adding the once keyword after a file name you can eliminate all other copies of the map except the one at (0,0) to (1,1).
  image_map {

    gif "mypic.gif"

    once

  }

The map_type Option
The default projection of the image onto the x-y-plane is called a planar map type. This option may be changed by adding the map_type keyword followed by an integer number specifying the way to wrap the image around the object. 

A map_type 0 gives the default planar mapping.  The image is mapped onto the x-y-plane. The image is projected onto the object as though there were a slide projector somewhere in the -z-direction. The image exactly fills the square area from (x,y) coordinates (0,0) to (1,1) regardless of the image's original size in pixels. If you would like to change this default you may translate, rotate or scale the pigment or texture to map it onto the object's surface as desired.
A map_type 1 gives a spherical mapping. It assumes that the object is a sphere of any size sitting at the origin. The y-axis is the north/south pole of the spherical mapping. The top and bottom edges of the image just touch the pole regardless of any scaling. The left edge of the image begins at the positive x-axis and wraps the image around the sphere from west to east in a -y-rotation. The image covers the sphere exactly once. The once keyword has no meaning for this mapping type. 

With map_type 2 you get a cylindrical mapping. It assumes that a cylinder of any diameter lies along the y-axis. The image wraps around the cylinder just like the spherical map but the image remains one unit tall from y=0 to y=1. This band of color is repeated at all heights unless the once keyword is applied. 

Finally map_type 5 is a torus or donut shaped mapping. It assumes that a torus of major radius one sits at the origin in the x-z-plane. The image is wrapped around similar to spherical or cylindrical maps. However the top and bottom edges of the map wrap over and under the torus where they meet each other on the inner rim. 

//image map example 

// create a box that is twice as long as it is wide, height is 1/10 width

box {

    <-1.0, -0.5, -0.1>

    <1.0, 0.5, 0.1>

    texture {

       pigment {      

          image_map {                 // images are mapped (0,0) to (1,1)
             jpeg "bitmaps/candyland2-1.jpg"   // this creates a problem
          } 

          // scale <2,1,1>            // can be fixed with scale and translate
          // translate <1,0.5,0>      // of image
       } 

       finish {ambient 0.5}

    }     

    rotate <85,20,0>  

    scale 4
}   
//image map example, can be fixed with scale of object
box {

    <-0.5, -0.5, -0.1>

    <0.5, 0.5, 0.1>

    texture {

       pigment {      

          image_map {

             jpeg "bitmaps/candyland2-1.jpg"

          } 

          translate <0.5,0.5,0>

       } 

       finish {ambient 0.5}

    }     

    rotate <85,20,0>  

    scale <8,4,4>

}   

